LISTING 2. Program the Beginning Melody of The Star-Spangled Banner

;Chip Music Composing Example for the 8051-like Microcontroller

;Anthem.ASM: American National Anthem program for AT89C1051/2051/4051

;use 12 MHz XTAL or Oscillator, 1 Cycle = 1 us 

SPKR   BIT   P1.7                       ; Speaker connected to P1.7 (pin 19)

TICK   DATA  10H                        ; Timer0 Tick Counter

;----------------------------------------------------------------------------

     ORG    0000H

          AJMP    INIT

     ORG    000BH                       ; Timer0 Overflow Interrupt Vector

          AJMP    TIMOVF                ; Interrupt Service routine; 2us

INIT:

          MOV     A, #30H

          MOV     SP, A                 ; set up stack pointer

          MOV     IE, #82H              ; Enable Timer0 Interrupt

          MOV     TMOD, #03H            ; Timer0 work in mode 3: 8-bit Timer

          SETB    TR0                   ; turn ON Timer0

REPET:

        call   Sew

        call   Sew

        call   Me

        call   Doe

        call   Doe

        call   Me

        call   Me

        call   Sew

        call   Sew

        call   Doo

        call   Doo

        call   Doo

        call   Doo

        call   Doo

        call   Doo

        call   Doo

        call   Doo

        call   DLhalfS

        call   DLhalfS

        call   DLhalfS

        call   DLhalfS

        ajmp   REPET

;------------------------------------

;This routine delays 10 ms at 12 MHz

DLY10ms:

        PUSH    ACC

        PUSH    B

        MOV     A, #227

LOOPN1:

        MOV     B, #20

LOOPN2:

        DJNZ    B, LOOPN2

        DEC     A

        JNZ     LOOPN1

        POP     B

        POP     ACC

        RET

;--------------------------

DL100ms:

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        ACALL   DLY10ms

        RET

;---------------------------

;delay .5 sec at 12 MHz 

DLhalfS:

        ACALL   DL100ms

        ACALL   DL100ms

        ACALL   DL100ms

        ACALL   DL100ms

        ACALL   DL100ms

        RET

;------------------------------------------------

; Music Note Subroutines

;------------------------------------------------

; 523 Hz "C5" tone, T/2 = 956 us

Doe:

C5:

        MOV     R7, #0

REPETCh:

        MOV     TICK, #0     ; clear Tick Counter

loopCh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #63, loopCh1

        MOV     TICK, #0     ; clear Tick Counter

loopCh2:

        SETB    SPKR         ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #63, loopCh2

        INC     R7

        MOV     A, R7

        CJNE    A, #128, REPETCh

        RET

;--------------------------------------------------

; 587 Hz "D5" tone, T/2 = 852 us

Ray:

D5:

        MOV     R7, #0

REPETDh:

        MOV     TICK, #0     ; clear Tick Counter

loopDh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #56, loopDh1

        MOV     TICK, #0     ; clear Tick Counter

loopDh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #56, loopDh2

        INC     R7

        MOV     A, R7

        CJNE    A, #143, REPETDh

        RET

;--------------------------------------------------

; 659 Hz "E5" tone, T/2 = 759 us

Me:

E5:

        MOV     R7, #0

REPETEh:

        MOV     TICK, #0     ; clear Tick Counter

loopEh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #50, loopEh1

        MOV     TICK, #0     ; clear Tick Counter

loopEh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #50, loopEh2

        INC     R7

        MOV     A, R7

        CJNE    A, #161, REPETEh

        RET

;--------------------------------------------------

; 698 Hz "F5" tone, T/2 = 716 us

Far:

F5:

        MOV     R7, #0

REPETFh:

        MOV     TICK, #0     ; clear Tick Counter

loopFh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #47, loopFh1

        MOV     TICK, #0     ; clear Tick Counter

loopFh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #47, loopFh2

        INC     R7

        MOV     A, R7

        CJNE    A, #170, REPETFh

        RET

;--------------------------------------------------

; 784 Hz "G5" tone, T/2 = 638 us

Sew:

G5:

        MOV     R7, #0

REPETGh:

        MOV     TICK, #0     ; clear Tick Counter

loopGh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #42, loopGh1

        MOV     TICK, #0     ; clear Tick Counter

loopGh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #41, loopGh2

        INC     R7

        MOV     A, R7

        CJNE    A, #191, REPETGh  

        RET

;--------------------------------------------------

; 880 Hz "A5" tone, T/2 = 568 us

La:

A5:

        MOV     R7, #0

REPETAh:

        MOV     TICK, #0     ; clear Tick Counter

loopAh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #37, loopAh1

        MOV     TICK, #0     ; clear Tick Counter

loopAh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #37, loopAh2

        INC     R7

        MOV     A, R7

        CJNE    A, #214, REPETAh

        RET

;---------------------------------------------------

; 988 Hz "B5" tone, T/2 = 506 us

Tea:

B5:

        MOV     R7, #0

REPETBh:

        MOV     TICK, #0     ; clear Tick Counter

loopBh1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #33, loopBh1

        MOV     TICK, #0     ; clear Tick Counter

loopBh2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #33, loopBh2

        INC     R7

        MOV     A, R7

        CJNE    A, #241, REPETBh

        RET

;---------------------------------------------------

; 1047 Hz "C6" tone, T/2 = 478 us

Doo:

C6:

        MOV     R7, #0

REPETCs:

        MOV     TICK, #0     ; clear Tick Counter

loopCs1:

        CLR     SPKR         ; SPKR=LOW

        MOV     A, TICK

        CJNE    A, #31, loopCs1

        MOV     TICK, #0     ; clear Tick Counter

loopCs2:

        SETB     SPKR        ; SPKR=HIGH

        MOV     A, TICK

        CJNE    A, #31, loopCs2

        INC     R7

        MOV     A, R7

        CJNE    A, #255, REPETCs

        RET

;------*--------*---------*-------*-------*----------------------

;Timer0 Overflow Interrupt Service Routine

;----------------------------------------------------------------

TIMOVF:

        MOV     TL0, #248    ; load Timer0 with 256-8=248; 2us

        inc     TICK         ; inc Tick Count by 1       ; 1us

        RETI                                             ; 2us

;--------------------------------------------------------------

; Total IET = 7 us/Interrupt

        END                               

